



INSTITUTO DE INVESTIGACIÓN FIIS-UNI
Facultad de Ingeniería Industrial y de Sistemas

TALLER DE PROGRAMACION CON ORACLE PL/SQL

Capítulo 04 Fundamentos de PL/SQL

Contenido

Introducción a PL/SQL

Tipos de Datos y Variables

Introducción a PL/SQL

Bloque

Unidad básica de programación.

Bloque anónimo

Sintaxis

```
Declare
    -----
    -----
Begin
    -----
    -----
End;
```

Script 1

```
Declare
    sFecha Varchar2(40);
Begin
    select to_char(sysdate,'dd/mm/yyyy hh24:mm:ss')
        into sFecha from dual;
    dbms_output.put_line( 'Hoy es: ' || sFecha );
End;
```

Función

Sintaxis

```
Create Or Replace Function NombreFuncion( Parámetros ) Return
TipoSalida
Is
    -----
    -----
Begin
    -----
    -----
End;
```

Script 2

```
create or replace function fnSuma( a number, b number ) return number
is
    c number;
begin
    c := a + b;
    return c;
end;
```

Ejecución:

```
SQL> select fnSuma(12,25) from dual;

FNSUMA(12,25)
-----
          37
```

Procedimiento

Sintaxis

```
Create Or Replace Procedure NombreProcedimiento( Parámetros )  
Is  
    -----  
    -----  
Begin  
    -----  
    -----  
End;
```

Script 3

```
create or replace procedure prSuma( a number, b number )  
is  
    c number;  
begin  
    c := a + b;  
    dbms_output.put_line( c );  
end;
```

Ejecución:

```
SQL> begin  
2     prSuma(15,15);  
3 end;  
4 /  
30  
  
PL/SQL procedure successfully completed.
```

Tipos de Datos y Variables

Tipos de Datos

Clase	Tipos de Datos
Escalares	Long, Number, Date, Char, Varchar2, Boolean
Compuestos	Record, Table
LOB	BFile, CLob, BLob
Referencia	Ref Cursor

Declaración de Variables

```
Nombre_Variable Tipo [CONSTANT] [NOT NULL] [:=valor];
```

Script 4

```
Declare  
  v_Descripcion varchar2(50);  
  v_NroAsiento  number := 35;  
  v_Contador    binary_integer := 0;
```

Asignar Valores a Variables

Caso 1: Sentencia de asignación

Script 5

```
v_NroAsiento := 78;  
v_Descuento := fnSuma(18,16);
```

Caso 2: Utilizando la sentencia SELECT

Script 6

```
select count(1) into v_Emps from emp;
```

Script 7

Procedimiento para consultar el nombre de un empleado.

```
create or replace procedure pr101(p_empno number)
is
  v_ename varchar2(10);
begin
  select ename into v_ename from emp where empno = p_empno;
  dbms_output.put_line(v_ename);
end;
```

Ejecución:

```
SQL> execute pr101(7499);
ALLEN

PL/SQL procedure successfully completed.
```

%Type

Permite declarar variables del mismo tipo de una columna.

Script 8

Función para consultar el nombre de un departamento.

```
create or replace function fn101(p_deptno dept.deptno%type)
return dept.dname%type
is
  v_dname dept.dname%type;
begin
  select dname into v_dname from dept where deptno = p_deptno;
  return(v_dname);
end;
```

Ejecución:

```
SQL> select fn101(10) from dual;

FN101(10)
-----
ACCOUNTING
```